


Modeling Example: A PAL

- Goal is to model a standard PAL
 - Both functionality and timing
- Will look at PAL16L8 model, but approach is valid for any standard PAL
- Functionality is defined via a JEDEC file
- Timing is defined via datasheets
- This model was written by Vince Sanders, MSU.

1/27/2003

BR

1



PAL 16L8

10 input only


1 output only

6 input/outputs

1/27/2003

BR

2



A Sample PAL

BR

3

Comments on Sample PAL


- 11 inputs, 3 outputs
 - Can implement three functions - functions can share inputs or not share inputs
- Each output implements a SOP equation with Four product terms.
 - Each product term can include complemented or uncomplemented form of an input.

1/27/2003

BR

4

Understanding the Diagram



Vertical Lines indicate a **product term**. Horizontal lines provide True and Complemented forms of **external** inputs.

Even though a product term looks like it has only one input, it actually has $2 * N$ inputs, where N is the number of external inputs.

1/27/2003


BR

5

Product Term



This looks like an AND gate with one input. Is actually:




Only drawn with a single line to save space.

1/27/2003

BR

6

Fuse Points



A cross over of a vertical input line and a horizontal product term line is a **FUSE LOCATION**. When the PAL is in its blank or erased state, all FUSES are connected. This means that each product term implements the equation:


(A A' B B' C C' K K') will be '0'. This means that the output will be high!

1/27/2003

BR

7

Programming



To program, will want to BLOW most of the fuses (break the vertical/horizontal crossover connection). To indicate a logic function, will use a 'X' over a fuse that I want to KEEP INTACT.

X ← Will mark Intact fuse location.


When a fuse is blown, that product term input acts as a '1' so that the input no longer effects the product term.

1/27/2003

BR

8

$$P' = D + J'$$



When implementing an equation, sometimes will not want to use all available product terms. If ALL fuses along product term are left intact, then product term value will be '0' and will not affect equation. Mark unused PT's by placing an X over them -- all fuses in that PT row are assumed intact.

Note that P' must be implemented!

1/27/2003


BR

9

Example Product Term AC'H'



The connections will be:




Actually, fuses are not 'blown' in erasable PLDs - the connection is broken in a non-destructive way, for erasable PLDs.

1/27/2003

BR

10

Another Example



$$P' = A'BGH' + CD' + HIJ + BG'H$$


1/27/2003

BR

11

JEDEC Files

- Model must read a JEDEC file that defines the programming
- Each fuse location is identified by a NUMBER



Fuse 0

Fuse 24


Just need to know if the fuse is intact or not. All fuses are intact in unprogrammed state.

1/27/2003

BR

12

Product Term Modeling Approach



Use a *resolved* data type for product term (column signals are simply multiple drivers on the product term)

Read fuse map

```
CONSTANT FuseMap : Bit_Vector := ReadJedec(JedecFileName);
CONSTANT rows : Natural := 64;
CONSTANT columns : Natural := 32;
CONSTANT outputs : Natural := 8;

SUBTYPE ResolvedAndSLV IS Resolve_AND Std_Logic;
TYPE ResolvedAndSLV IS ARRAY (Natural RANGE <=>) OF ResolvedAndSL;

SIGNAL AndTermsResolved : ResolvedAndSLV(0 TO rows - 1) := (OTHERS => '1'); --And Terms Resolved
SIGNAL AndTerms : Std_Logic_Vector(0 TO rows - 1); --And Terms
SIGNAL ci : Std_Logic_Vector(0 TO columns - 1); --Column Inputs
SIGNAL OrTerms : Std_Logic_Vector(0 TO outputs - 1); --Or Terms
SIGNAL fb : Std_Logic_Vector(1 TO 6); --fb(1 TO 6) <=> io(7 DOWNTO 2)
```

Model is for pal16L8 – fuse locations, # inputs, etc are all dependent on this PAL type.

1/27/2003

BR

19

Reading the JEDEC File

- Fuse Map read at ELABORATION time
 - I.e., elaboration time is when processes/signals that represent the VHDL model are created in-memory
- This is needed because VHDL GENERATE statements that are used to dynamically create signals/processes use information from the fuse map.
 - CANNOT read the fuse map at simulation time because this means the simulator has started and all processes/signals have already been created.
- Reading a file at elaboration time is difficult to debug because debugger is only available after elaboration time.

1/27/2003

BR

20

Resolution Function for Product Term Types

--Resolve_AND (Internal)

```
FUNCTION Resolve_AND (v : Std_Logic_Vector) RETURN Std_Logic IS
    VARIABLE result : Std_Logic := '1';
    BEGIN
        FOR ii IN v'RANGE LOOP
            result := result AND v(ii);
            EXIT WHEN result = '0';
        END LOOP;
        RETURN result;
    END Resolve_AND;
```

Note early exit when function is zero.

```
SUBTYPE ResolvedAndSLV IS Resolve_AND Std_Logic;
TYPE ResolvedAndSLV IS ARRAY (Natural RANGE <=>) OF ResolvedAndSL;
```

```
SIGNAL AndTermsResolved : ResolvedAndSLV(0 TO rows - 1) := (OTHERS => '1'); --And Terms Resolved
```

Note that this is a subtype of Std_Logic, which is itself a resolved type!!!!

1/27/2003

BR

21

Connecting Inputs to Column Signals

```
ColumnConnect_i1to8Gen: --(1 TO 8)
FOR ii IN 1 TO 8 GENERATE
    ci((ii-1)*4) <= TRANSPORT To_UX01(ii) AFTER WD_i(ii);
    ci((ii-1)*4 + 1) <= TRANSPORT NOT(ii) AFTER WD_i(ii);
END GENERATE ColumnConnect_i1to8Gen;
```

inverter

- Input signals are array *i*(1 to 8)
- column signals are *ci*(0 to num_columns - 1)
- Each input connected to a pair of column signals (2nd connection is a complemented version of the input)
- wd_i* are wire delay generics that are defined on the entity
- The GENERATE statement causes these signal assignments to be expanded at elaboration time

1/27/2003

BR

22

Product Term Connections

ColumnConnect_i1to8Gen: --(1 TO 8)

--And Plane

```
AndPlaneGen:
FOR row IN AndTermsResolved'RANGE GENERATE
    RowAllConnectedGen:
        IF (RowAllConnected(row)) GENERATE
            AndTermsResolved(row) <= '0';
        END GENERATE RowAllConnectedGen;
        RowNotAllConnectedGen:
            IF (NOT RowAllConnected(row)) GENERATE
                ColumnGen:
                FOR col IN ci'RANGE GENERATE
                    ConnectGen:
                        IF (FuseMap(row * ci'LENGTH + col) = connected) GENERATE
                            AndTermsResolved(row) <= ci(col);
                        END GENERATE ConnectGen;
                    END GENERATE ColumnGen;
                END GENERATE RowNotAllConnectedGen;
            END GENERATE AndPlaneGen;
```

Reduces # of signal assignments, improves performance

Only generated if fuse map location = '0'!!

1/27/2003

BR

23

RowAllConnected Function

Recall that if all column inputs are connected to a product term, then product term output is '0'. Can reduce model complexity (memory and execution time) if detect this case.

--RowAllConnected (Internal)

```
FUNCTION RowAllConnected (row : Natural) RETURN Boolean IS
    BEGIN
        RETURN NOT To_Boolean(Reduce_OR(
            FuseMap(row*columns TO (row+1)*columns - 1)));
    END RowAllConnected;
```

Reduce_OR function defined in VHDL for bit_vectors – returns a '1' if any bit in bit vector is a '1', else returns '0'.

1/27/2003

BR

24

Model Complexity

- Model complexity is proportional to memory required and execution time
- The number of signals, signal assignments, and processes in a model impacts complexity
 - More signals and signal assignments means more events means more execution effort required
 - More signals means more memory needed to track waveform history
- Use of GENERATE statements only creates the required signal assignments for the product terms based upon the fuse map contents

1/27/2003

BR

25

OR Terms

std_logic_vector

ResolvedAndSLV type

Can do 'type cast' without explicit conversion function because ResolvedAndSL is subtype of Std_logic.

~Now AndTerms gets resolved signal
AndTerms <= Std_Logic_Vector(AndTermsResolved);

~Or Plane

OrPlaneGen:
FOR ii IN OrTerms'RANGE GENERATE
OrTerms(ii) <= Reduce_OR(AndTerms((ii*8 + 1) TO (ii*8 + 7)));
END GENERATE OrPlaneGen;

Reduce_OR function defined for std_logic_vector in 1164 standard.

1/27/2003

BR

26

```
TriStates_1to6Gen:
FOR ii IN 1 TO 6 GENERATE
PROCESS (OrTerms(ii), AndTerms(ii*8))
  ALIAS i : Std_Logic IS OrTerms(ii);
  ALIAS oe : Std_Logic IS AndTerms(ii*8);
  ALIAS o : Std_Logic IS io(8-ii);
  ALIAS LD_o : Time IS LD_io(8-ii);
BEGIN
CASE oe IS
WHEN '0' => --TriStated Output
  IF (oe'EVENT) THEN
    o <= TRANSPORT 'X' AFTER 0 NS,
      'Z' AFTER (t.tcr + LD_o);
  END IF;
WHEN '1' => --Output Enabled (NOTE INVERTED)
  IF (oe'EVENT) THEN
    o <= TRANSPORT 'X' AFTER 0 NS,
      NOT i AFTER (t.tea + LD_o);
  ELSE
    IF ( (t.tpd + LD_o) < (t.tea + LD_o - oe'LAST_EVENT) ) THEN
      o <= TRANSPORT NOT i AFTER (t.tea + LD_o - oe'LAST_EVENT);
    ELSE
      o <= TRANSPORT NOT i AFTER (t.tpd + LD_o);
    END IF;
  END IF;
WHEN OTHERS => o <= TRANSPORT oe AFTER tpdX;
END CASE;
END PROCESS;
END GENERATE TriStates_1to6Gen;
```

Tri-State Buffers

GENERATE statements
can be used for processes

ALIAS statements used to
improve readability